
ABSTRACT

     PROC REPORT started as a souped-up version of PROC PRINT.  Now this unique product combines features from PROC PRINT, SORT, FREQ, MEANS, and TABULATE.  Because of its special blend of possibilities, PROC REPORT is often the easiest way to do an elegant data listing or a report with descriptive statistics.

    This tutorial shows how to use the batch versions of PROC REPORT.  Via examples, it thoroughly explores the use of PROC REPORT in creating specialized data listings.

INTRODUCTION

     Many people shy away from PROC REPORT because of it’s mysterious defaults.  Or put another way, PROC REPORT acts differently than other SAS( PROC’s. However, once you see how PROC REPORT works, you may never want to go back to other PROC’s. 

    PROC REPORT does the yeoman’s work of PROC PRINT, SORT, FREQ, MEANS, and TABULATE, and PUT-Statement Formatting (DATA _NULL_) all in one procedure.

    The purpose of this paper is to examine, in detail, the how to produce listings with batch-mode PROC REPORT.  That is, this paper will look at the PRINT, SORT and DATA _NULL_ aspects of PROC REPORT.  The FREQ, MEANS and TABULATE aspects will be examined in a subsequent article.  All of the examples in this paper are designed to function using SAS version 6.12 or higher.

     This tutorial is intended for SAS programmers with knowledge of the SAS dataset structure, and exposure to the SAS PRINT and CONTENTS procedures.  After completing this article, the reader should be able to do data listing using PROC REPORT and have the foundation for going on to learn how to do descriptive statistics with the procedure.

Sample Input Data Set

    The following two Exhibits are a PROC PRINT and a partial PROC CONTENTS of an example data set. This data set is going to be used as input to all the examples of PROC REPORT presented in this paper.

Exhibit 1. Listing of Input Data Set

------------------------------------

PROC PRINT DATA=ORG NOOBS UNIFORM;

RUN;

------------------------------------

      OUTPUT... file=ORG 

------------------------------------

 ID    SITE  DUM  NAME  SEX  AGE 

 A01    RU    1   SUE    F    58  

 A02    LA    1    X     M    58  

 A04    RU    1   TOM    M    21  

 A07    LA    1   LEE    F    47  

 A08    LA    1   KAY    F    29  

 A10    RU    1          M    36  

------------------------------------

Exhibit 2.  Partial CONTENTS of Data Set

----------------------------------------

   Variables Ordered by Position

----------------------------------------

 # Variable  Type  Len  Pos  Label

 1   ID      Char    4    0

 2   SITE    Char    2    4

 3   DUM     Num     8    6

 4   NAME    Char    3   14

 5   SEX     Char    1   17

 6   AGE     Num     8   18  Age in years

----------------------------------------

Features & Defaults of Proc REPORT

    Observe Exhibit 3. It shows a very simple PROC REPORT.  This example illustrates the procedure’s defaults. 

Exhibit 3. REPORT’s defaults in Batch Mode

------------------------------------------

     PROC REPORT DATA=ORG NOWINDOWS;

     RUN;

------------------------------------------

                            S

        SI             NAM  E     Age in

  ID    TE        DUM  E    X      years

  A01   RU          1  SUE  F         58

  A02   LA          1   X   M         58

  A04   RU          1  TOM  M         21

  A07   LA          1  LEE  F         47

  A08   LA          1  KAY  F         29

  A10   RU          1       M         36

------------------------------------------

Exhibit 4. Variable Names as Col Headings

------------------------------------------

     OPTIONS NOLABEL;

     PROC REPORT DATA=ORG NOWINDOWS;

     RUN;

------------------------------------------

                            S

        SI             NAM  E     

  ID    TE        DUM  E    X        AGE

  A01   RU          1  SUE  F         58

  A02   LA          1   X   M         58

  A04   RU          1  TOM  M         21

  A07   LA          1  LEE  F         47

  A08   LA          1  KAY  F         29

  A10   RU          1       M         36

------------------------------------------

Exhibit 5. Unexpected Results

------------------------------

PROC REPORT

      NOWINOWS

      DATA=ORG (KEEP=DUM AGE) 

    ;

RUN;

------------------------------

       OUTPUT

------------------------------

           Age in

     DUM    years

      6       249

------------------------------

The Define Statement

Exhibit 6. The DEFINE statement and NOWD

------------------------------------------

PROC REPORT DATA=ORG (Keep=DUM AGE) NOWD ;

       DEFINE  DUM  /DISPLAY; 

       DEFINE  AGE  /DISPLAY; 

RUN;

------------------------------------------

               Age in

       DUM      years

         1         58

         1         58

         1         21

         1         47

         1         29

         1         36

------------------------------------------

Exhibit 7. Ordering/Subsetting Cols/Vars

----------------------------------------

   PROC REPORT   NOWD   DATA=ORG  ;

       COLUMN  AGE  DUM;

       DEFINE  AGE  /DISPLAY;

       DEFINE  DUM  /DISPLAY; 

   RUN;

----------------------------------------

         Age in         

          Years     DUM

             58      1 

             58      1 

             21      1 

             47      1 

             29      1 

             36      1 

----------------------------------------

Exhibit 8. PROC REPORT’s Syntax So Far

------------------------------------------

OPTIONS NOLABEL;

PROC REPORT

          NOWD

          DATA=file-name

            (OBS= WHERE=() DROP= KEEP=)

        ;

    COLUMN list-of-variables;      

    DEFINE  var-name /DISPLAY;

    DEFINE  var-name /DISPLAY;

RUN;

------------------------------------------

HORizontal SPACING and more

Ehibit 9. A Proc-PRINT Type of Listing

------------------------------------------

PROC REPORT NOWD

             DATA=ORG (WHERE=(ID<"A08")) ;

    COL ID SITE NAME AGE ;

    DEFINE ID/DISPLAY;

    DEFINE SITE/DISPLAY;

    DEFINE NAME/DISPLAY;

    DEFINE AGE/DISPLAY;

RUN;

------------------------------------------

             SI  NAM     Age in

       ID    TE  E        years

       A01   RU  SUE         58

       A02   LA   X          58

       A04   RU  TOM         21

       A07   LA  LEE         47

------------------------------------------

DEFAULT HORIZONTAL SPACING

· Default Spacing between fields = 2 blanks

· Default Justification (=Alignment)

· RIGHT for Numeric Fields

· LEFT for Character Fields

· If no format specified, then Proc REPORT uses

· Best9.  for Numeric Fields

· $w. for Character Fields (w=width)

· Default Width= Format width

Exhibit 10.Cleaning Up Exhibit 9’s Listing

------------------------------------------

PROC REPORT HEADLINE HEADSKIP NOWD

             DATA=ORG(WHERE=(ID<"A07")); 

    COL ID SITE NAME AGE ;

    DEFINE ID   /DISPLAY;

    DEFINE SITE /DISPLAY WIDTH=4 RIGHT;

    DEFINE NAME /DISPLAY WIDTH=4 RIGHT;

    DEFINE AGE  /DISPLAY "AGE" WIDTH=3; RUN;

------------------------------------------

      ID    SITE  NAME  AGE

      A01     RU   SUE   58

      A02     LA     X   58

      A04     RU   TOM   21

------------------------------------------

Exhibit 11. Label Hierarchy

------------------------------------------

REPORT uses the 1st label it finds in this list.

· col-heading”  (DEFINE option)

· OPTIONS NOLABEL;  (implies Var Names)

· LABEL Statement in Proc REPORT

· LABEL in the Data Descriptor

· Variable Names

------------------------------------------

Exhibit 12. Some DEFINE Statement Options

-----------------------------------------

· DISPLAY

· Col-Heading in quotes

· SPACING= (Overrides all Spacings)

· FORMAT= (Overrides all other Formats)

· WIDTH=  (Default=Format Width)

· Justification Specifications

· RIGHT

· LEFT

· CENTER

Exhibit 13. Justification Rules

   (Right, Left, Center)

------------------------------------------

  Justification applies to

· col-headings

· data values.

  Default Justification (=Alignment)

· RIGHT for Numeric Fields

· LEFT for Character Fields

  Numerical values: 

· always remain right justified within FORMATs

· FORMATs are justified within the WIDTH.

  For col-headings & character values: 

· values are justified within the WIDTH.  (without regard for the $w. FORMAT).

· leading blanks are retained.

· trailing blanks are eliminated.

------------------------------------------

Exhibit 14. RIGHT Justification 

------------------------------------------

DEFINE AGE / ‘AGE(’ SPACING=3 WIDTH=6

              FORMAT=4. RIGHT;

------------------------------------------

COL=  123456789        s=Spacing location

TYPE= ssswwwwww        w=Width location

TYPE=      ffff        f=Format location

            AGE        (=Blank space

           ((58

------------------------------------------

Exhibit 15. LEFT Justification

------------------------------------------

DEFINE AGE / ‘(AGE’” SPACING=3 WIDTH=6

              FORMAT=4. LEFT;

------------------------------------------

COL=  123456789        s=Spacing location

TYPE= ssswwwwww        w=Width location

TYPE=    ffff          f=Format location

         (AGE          (=Blank space

         ((58

------------------------------------------

Exhibit 16. Forcing Trailing Blanks

------------------------------------------

DATA _NULL_;

   CALL SYMPUT('blk','FF'X);

RUN;

DEFINE AGE / ( AGE&blk( SPACING=3 WIDTH=6

              FORMAT=4. RIGHT;

------------------------------------------

COL=  123456789        s=Spacing location

TYPE= ssswwwwww        w=Width location

TYPE=      ffff        f=Format location

           AGE(
           ((58

------------------------------------------

Exhibit 17. The Flow Option

------------------------------------------

PROC REPORT Headline NOWD DATA=ORG; 

    COL ID NAME AGE NOTE;

    DEFINE NAME /DISPLAY WIDTH=4 RIGHT;

    DEFINE AGE  /DISPLAY "AGE" WIDTH=3;

    DEFINE NOTE /DISPLAY WIDTH=13 FLOW;

------------------------------------------

     ID    NAME  AGE  NOTE         

     A01    SUE   58  This is an ex

                      of FLOW.

     A02      X   58  No flow here.

     A04    TOM   21  Adverse

                      Event.

------------------------------------------

Exhibit 18. Expanding dash.

            Splitting slash.
------------------------------------------

PROC REPORT Headline NOWD DATA=ORG; 

  COL ID NAME AGE NOTE;

  DEFINE NAME /DISPLAY WIDTH=7 RIGHT

             ‘First/Name”’;

  DEFINE AGE  /DISPLAY "AGE" WIDTH=3;

  DEFINE NOTE /DISPLAY WIDTH=13 FLOW

            ‘- Note- ‘; RUN;

------------------------------------------
            First

     ID      Name  AGE  ---- NOTE ---

     A01      SUE   58  This is an ex

                        of FLOW.

------------------------------------------
Exhibit 19. Expanding star.

            Split= option.

------------------------------------------

PROC REPORT Headline NOWD DATA=ORG

        SPLIT=’*’;

    COL ID NAME AGE NOTE;

    DEFINE NAME /DISPLAY WIDTH=7 RIGHT

             ‘First*Name’ 

    DEFINE AGE  /DISPLAY "AGE" WIDTH=3;

    DEFINE NOTE /DISPLAY WIDTH=13 FLOW

            ‘* Note *’; RUN;

------------------------------------------
            First 

     ID      Name  AGE  **** NOTE ***

     A01      SUE   58  This is an ex

                        of FLOW.

------------------------------------------
Exhibit 20. Special Column Heading

------------------------------------------

PROC REPORT Headline NOWD DATA=ORG;

    COL (‘-ID INFO-‘ ID NAME) AGE NOTE;

    DEFINE ID   /DISPLAY ‘ID#’;

    DEFINE NAME /DISPLAY WIDTH=6 RIGHT;

    DEFINE AGE  /DISPLAY  WIDTH=3;

    DEFINE NOTE /DISPLAY WIDTH=13 FLOW

            ‘* Note *’;    RUN;

------------------------------------------
     --ID INFO---

     ID#     NAME  AGE  **** NOTE ***

     A01      SUE   58  This is an ex

                        of FLOW.

------------------------------------------

Exhibit 21. Overlining the column headings

------------------------------------------

PROC REPORT Headline NOWD DATA=ORG;

   COL (‘_ _’(‘-ID INFO-‘ ID NAME) Age 

                                 Note);

    DEFINE ID / ‘ID#’;

    DEFINE NAME /DISPLAY WIDTH=6 RIGHT;

    DEFINE AGE  /DISPLAY  WIDTH=3;

    DEFINE NOTE /DISPLAY WIDTH=13 FLOW

            ‘< Note >’;    RUN;

------------------------------------------
                                     

     --ID INFO---

     ID#     NAME  AGE  <<<< NOTE >>>

     A01      SUE   58  This is an ex

                        of FLOW.

------------------------------------------

Exhibit 22.Summary of Column Headings

------------------------------------------

· Controlling the Breaks (stacking text)

· ‘xxx’ ‘yyy’ (multiple pairs of  quotes)

· ‘xxx/yyy’ (slash is default split char.)

· ‘xxx*yyy’ (with SPLIT=’*’ Report Option)

· Underlining

 HEADLINE Proc Report Statement Option

· Overlining (with text or characters)

          COL (‘text’ var var) var var;

· Extending Headers to Column Width

· ‘char header-text char’

          (where char pair is **  ==  ++ 

                    --  _ _  ..  < >   > <   )

· Justification (LEFT, CENTER, RIGHT)

------------------------------------------

why people do use proc report

    Complete Horizontal Control

· Complete Col-heading Control

· Justification Control

· The FLOW option

· More Professional Looking Reports

Exhibit 23. The ORDER Option.

----------------------------------------

  Proc REPORT nowd headline headskip

              Data=ORG; 

    COL SITE NAME AGE ;

    DEFINE SITE /ORDER WIDTH=4;

    DEFINE NAME /DISPLAY WIDTH=4;

    DEFINE AGE  /DISPLAY "AGE" WIDTH=3; 

----------------------------------------

          SITE  NAME  AGE

           LA     X     58

                 LEE    47

                 KAY    29

           RU    SUE    58

                 TOM    21

                        36

----------------------------------------

Exhibit 24. Sort on two variables 

------------------------------------------

  Proc REPORT nowd headline headskip 

              Data=ORG; 

    COL SITE NAME AGE ;

    DEFINE SITE /ORDER WIDTH=4;

    DEFINE NAME /ORDER WIDTH=4;

    DEFINE AGE  /DISPLAY "AGE" WIDTH=3; 

------------------------------------------
          SITE  NAME  AGE

           LA     X     58

                 KAY    29

                 LEE    47

           RU    SUE    58

                 TOM    21

------------------------------------------
Exhibit 25. MISSING Option

------------------------------------------

  Proc REPORT nowd headline headskip 

              Data=ORG MISSING; 

    COL SITE NAME AGE ;

    DEFINE SITE /ORDER WIDTH=4;

    DEFINE NAME /ORDER WIDTH=4;

    DEFINE AGE  /DISPLAY "AGE" WIDTH=3; 

------------------------------------------
          SITE  NAME  AGE

           LA     X     58

                 KAY    29

                 LEE    47

           RU           36

                 SUE    58

                 TOM    21

------------------------------------------
 

Exhibit 26. General SAS Statements.

----------------------------------------

    Supported by PROC REPORT.

· WHERE

· LABEL - not recommended

· FORMAT- not recommended

· TITLE

· FOOTNOTE

· BY

   NOT supported by PROC REPORT

· KEEP

· DROP

----------------------------------------

Exhibit 27. Format Hierarchy  

----------------------------------------

REPORT accepts the 1st Format from the following list that fits in WIDTH.

(Thus, WIDTH can affect FORMAT/Values.)

· FORMAT=  (DEFINE option)

· FORMAT Statement in Proc REPORT

· FORMAT in the Data Descriptor

· Default Formats as follows

· Best9.  for Numeric Fields

· $w. for Character Fields (w=width)

· If Value does not fit in Format

· Numeric Fields are filled with *

· Character Fields are truncated

------------------------------------------

Exhibit 28. Order= Option                                         ------------------------------------------

  Proc REPORT nowd missing headline 

          headskip Data=ORG; 

    COL  SITE NAME AGE ;

    WHERE AGE>30. ;

    DEFINE SITE /ORDER WIDTH=14

          FORMAT=$SITE. ORDER=INTERNAL;

    DEFINE NAME /ORDER WIDTH=4;

    DEFINE AGE  /DISPLAY "AGE" WIDTH=3; 

------------------------------------------
      SITE            NAME  AGE

      LOS ANGELES      X     58

                      LEE    47

      DURHAM-RALEIGH         36

                      SUE    58

------------------------------------------

Exhibit 29. ORDER= Option  Values                           ------------------------------------------

ORDER= (one of the following)

    DATA or FORMATTED or FREQ or INTERNAL

Meaning of Options

· DATA - Order recs. as in the data set.

· FORMATTED - Sort values after formatting.

· INTERNAL - Sort values before formatting them.

· FREQ - Sort values by frequency of occurrence in the data set.

Defaults

· DATA when manipulation-type is DISPLAY.

· FORMATTED when manip.-type is ORDER.

· (in every other PROC, default is INTERNAL.)

Exhibit 30. BREAK  Statement  

------------------------------------------Proc REPORT nowd missing headline DATA=ORG

    COL  SITE NAME AGE ;

    WHERE AGE>30. ;

    DEFINE SITE /ORDER WIDTH=4;

    DEFINE NAME /ORDER WIDTH=4;

    DEFINE AGE  /DISPLAY "AGE" WIDTH=3; 

    BREAK AFTER SITE /SKIP;

RUN;

------------------------------------------
          SITE  NAME  AGE

           LA     X     58

                 LEE    47

           RU           36

                 SUE    58

------------------------------------------
Exhibit 31. Syntax for Listings

------------------------------------------

OPTIONS NOLABEL LS= PS=;

PROC REPORT   NOWD  MISSING SPLIT=’char’

      HEADLINE HEADSKIP 

      DATA=file-nam

        (OBS= Where=() Drop= Keep=);

  WHERE expression ;

  TITLE ‘your-message‘ ;

  FOOTNOTE ‘your-message’;

  COLUMN (‘(header&blk’ list-of-vars) 

             more-vars;      

  DEFINE  var-name /<DISPLAY or ORDER>

          ‘(col-head&blk”

           SPACING=  WIDTH=  FORMAT=  FLOW

           RIGHT  LEFT  CENTER 

  BREAK AFTER order-var /SKIP;

RUN;

